

Surviving Survival Analysis 101: Making the Likelihood Ratio Test Easier Using a SAS® MACRO

Katelyn J. Ware, Grand Valley State University and Spectrum Health Office of Research;
Rachel R. Baxter, Grand Valley State University and Spectrum Health Office of Research

ABSTRACT

When modeling survival analysis data, if a certain covariate does not meet the proportional hazards assumption, it can still be included in the Cox Proportional Hazards model with stratification. However, one must decide if the slopes of the other covariates are statistically different across strata or not. If they are, an interaction term must be included in the model.

The likelihood ratio test is a useful tool when deciding if interaction terms are needed in a stratified Cox Proportional Hazards model. The $-2\log L$ values may be obtained with PROC PHREG and used to compute a chi-square likelihood ratio test and p-value. The following paper describes a MACRO created in SAS® 9.4, which automates this test.

INTRODUCTION

As described by Kleinbaum and Klein (2012), the stratified Cox Proportional Hazards (PH) model is an adaptation of the Cox PH model. It allows for control of a predictor variable not satisfying the proportional hazards (PH) assumption through stratification, while simultaneously including covariates in the model that do satisfy the PH assumption. This model assumes no-interaction which means it assumes the parameter estimates do not vary over levels of the stratum. To test if the parameter estimates do vary, the likelihood ratio test is used to compare log-likelihood statistics for the interaction and no-interaction model. This method is commonly employed by statisticians to decide whether to use the interaction model or not.

The PHREG Procedure does not currently offer the likelihood ratio test as an output when looking at two stratified Cox PH models. We created a SAS® MACRO to fill this gap by allowing the user to obtain parameter estimates for both models, the likelihood ratio test statistic, and its corresponding p-value. This paper explains our SAS® MACRO that allows users to decide between the interaction and no-interaction stratified Cox PH models.

SURVIVAL ANALYSIS BACKGROUND

Several survival analysis references, such as the book written by Kleinbaum and Klein in 2012, describe how the statistical methods of survival analysis provide analysis of time to event data where the unit of observation (patient, participant, etc.) can be included in the study, regardless of whether they leave the study early or never experience the event. These incomplete observations are called right censored observations and are still included as they provide valuable information for the time they are in the study. Two important functions in modeling survival analysis data are the survival probability and hazard rate. The survival probability is the probability that a random individual survives (does not experience the event of interest) past a certain time (t). The hazard rate is the rate per unit of time that an individual experiences the event in the next unit of time given they have “survived” up until that time. The Cox PH model is the commonly used modeling technique for the above-mentioned survival analysis data. It models the hazard function, noted $h(t)$, with its relationship to any covariates of interest. The hazard function, $h(t)$, gives the instantaneous potential per unit time for the event to occur, given that the individual has survived up to time t . This is loosely described as risk. It is formally defined as

$$\lim_{\Delta t \rightarrow 0} \frac{P(t \leq T < t + \Delta t | T \geq t)}{\Delta t}$$

The Cox PH model differs from typical ordinary least squares regression due to the covariates predicting the hazard rate function rather than the response variable (t) itself. It is written as

$$h(t) = h_0(t)e^{\beta_1x_1 + \beta_2x_2 + \dots + \beta_px_p}$$

where $h_0(t)$ is the baseline hazard function, x_1, x_2, \dots, x_p are the p number of predictors, and $\beta_1, \beta_2, \dots, \beta_p$ are the parameter coefficients.

Throughout this paper we assume the typical conditions (i.e. specifically non-informative censoring and sufficiently large sample sizes) are met for likelihood based statistical inference for survival analysis. This model assumes the hazard rate for two different levels of a covariate are proportional across the duration of the study for each covariate included. For example, if women have twice the risk of a stroke at $t=4$ months in the study, they will also have twice the risk at any other time. This assumption can be assessed with various checks in SAS[®]. Gillespie (2006) and Yao (2018) describe several ways to assess the PH assumption with corresponding SAS[®] code. Here we briefly describe three of these methods.

For visual assessments, one can use graphical checks. The most well-known of which is the Kaplan Meier plot, commonly known as the *Log-log Plot* (Figure 1). This graph plots the Kaplan Meier curves of each of the categories for every covariate used in the model against $\ln(t)$. If the PH assumption is reasonable, the curves should be parallel; or if the curves do cross, they should not have any large separation between the curves before and after they cross (meaning they are essentially on top of each other).

Another graphical check is the *Observed vs. Expected Plot* (Figure 2). This graph checks the Kaplan Meier curves (which is an observed plot) and compares them to the unstratified adjusted survival curves. This latter plot is produced using the unstratified Cox PH model. If the two graphs look similar for each level of the covariates, then it is safe to conclude that the PH assumption is reasonable. With both the *Log-Log plot* and the *Observed vs. Expected plot* graphical checks, the quantitative variables must be categorized, and one should remember interpreting graphs can be subjective.

The Goodness of Fit Test is a null hypothesis significance test to assess assumptions. Its null hypothesis states that the PH assumption is met, meaning one hopes to fail to reject the null hypothesis. To perform the test, Schoenfeld residuals are used, which should be independent of time if the PH assumption is met. This means there should be no correlation between the residuals and time. The test provides a p-value for a correlation test. Since we do not want to reject the null hypothesis, a large p-value is desired. One difference between this check and the two listed above, aside from not being a visual check, is that a quantitative covariate should be used in its quantitative form and not categorized. Also, if the assumption is not met, there is no way of knowing how the PH assumption is violated, as it does not provide a way to visually assess it. We chose not to show an example of this test here because of spatial and visual reasons.

Figure 1 and Figure 2 display graphs based on an example data set, 'addicts', which is described in the MACRO section later in the paper (<http://web1.sph.emory.edu/dkleinb/surv3.htm#data>). Figure 1 is a *Log-Log plot* of a variable called clinic, and the PH assumption is not met because the lines are separating, and as such, are not deemed to be sufficiently parallel. In Figure 2, an example of the *Observed vs. Expected plot* for a binary variable called prison-demonstrates that the PH assumption is met because the observed line for each prison is reasonably close to the expected line for each prison.




Figure 1. Log-Log Plot of the variable clinic. Clinic is shown to be violating the Proportional Hazards Assumption due to the lines separating meaning they are not deemed sufficiently parallel.




Figure 2. Observed vs. Expected Plot of the variable prison. This binary variable, prison, demonstrates a scenario where the Proportional Hazards Assumptions is satisfied with the observed lines for prison 0 (blue) and prison 1 (green) are similar to the expected lines for prison 0 (red) and prison 1 (black).

STRATIFIED COX PROPORTIONAL HAZARDS MODEL

What if the hazard rates are not truly proportional across the levels of a variable, as in Figure 1 above? One option is to include the variable in the model, but not test for its effect. This is known as the stratified Cox PH model, where strata are defined by a covariate which does not satisfy the PH assumption. This model is also described in Chapter 5 of Kleinbaum and Klein (2012) and is best used when the covariate that fails the PH assumption has a small number of categories (or if it is quantitative, it can be categorized into a small number of levels) and is typically deemed to be a nuisance covariate. In this method, the effect of the covariate is no longer quantifiable, so it only makes sense if the covariate in question needs to be controlled for and no regression parameter is needed. In this stratified model, the baseline hazard functions will be different across the different stratum (as its hazard rate was not found to be proportional to the other covariates). However, the parameter estimates for the other covariates will be the same across stratum levels. In terms of the Cox PH model notation, $h_0(t)$ varies for each stratum level, but the coefficients ($\beta_1, \beta_2, \dots, \beta_p$) are the same across stratum. This is noted as

$$h_g(t) = h_{0g}(t)e^{\beta_1x_1 + \beta_2x_2 + \dots + \beta_px_p}$$

where g indicates the g^{th} stratum. Thus, there are g different equations, all with the same slopes but different hazard functions for the g levels of the stratifying variable.

What if we think the slopes are different for each level of the stratum? This would suggest that there is an interaction between the covariates and the stratum variable. Then, the slopes should be allowed to vary across the different stratum equations. This is similar to stratification in simple linear regression when an additional categorical covariate is added. In simple linear regression, if the slopes are forced to be the same across each level of the categorical covariate, then only the y-intercepts are different in each stratum equation. However, one must decide if different y-intercepts are enough or if the slopes too, depend on the level of the categorical covariate, which would mean one must allow the slopes to change as well. Adding an interaction term allows for the slopes to vary across levels. The necessity of determining whether there might be interaction between the covariates and the stratum variable will be shown in our application of the Cox PH model in our example data set.

LIKELIHOOD RATIO TEST

As with most modeling situations, it is advantageous to keep the model as simple as reasonably possible, so we do not want to include the two-way interactions between the stratifying variable and each independent covariate if they are not needed. We can formally test this using the likelihood ratio test, which is a hypothesis test used to decide between a hierarchical class of models in all applications of statistical modeling in addition to survival. The test uses a ratio comparison of maximum likelihood values to determine which model is the best (Kleinbaum et. al., 2009). According to Kleinbaum and his coauthors, the comparison is between various versions of the same model with differing parameter values involved, essentially nesting models within each other. The formula used to compute the test statistic when comparing to models is

$$LR = -2 \ln \left(\frac{L_R}{L_F} \right) = -2 \ln L_R - (-2 \ln L_F)$$

where L_R is the maximized likelihood for the reduced model and L_F is the maximized likelihood of the more complex model.

The test will help determine the best model by looking at the magnitude of the ratio of the two maximized likelihood values. This magnitude reflects how much the maximized likelihood value of the reduced model changes based on the addition of one or more parameters. Given that there is a sufficiently large sample, the likelihood ratio statistic will have an approximate chi-square distribution with degrees of freedom equal to the difference in the number of parameters between its two models under its null hypothesis. The null

hypothesis states that the extra (interaction) parameters in the full model are equal to 0. Computing the maximum likelihoods by hand can be lengthy and difficult. However, most software can now compute output providing the log-likelihood values which makes the calculation a much less daunting task.

In the stratified Cox PH model, the likelihood ratio test compares the interaction model with the no-interaction model. The no-interaction model is considered the nested or reduced model as it contains only the main effects of each covariate, while the interaction model contains the main effects of each independent variable and the two-way interactions between the stratum variable and each independent variable. Unlike in other regression settings, the main effect of the stratum variable is captured through the baseline hazard functions, so there is no regression parameter for the main effect of the stratum variable. This full model is noted as

$$h_g(t) = h_{0g}(t)e^{\beta_{1,g}x_1 + \beta_{2,g}x_2 + \dots + \beta_{p,g}x_p}$$

where each of the g equations have varying slopes.

As mentioned before, this can also be thought of in terms of an interaction between the stratum variable and each covariate. For example, consider modeling the hazard function with independent covariates gender (male and female), age (continuous), and stratum variable treatment (A, B, and C, respectively). As in all regression, indicator variables must be used for gender and treatment since they are categorical. In SAS®, including the categorical variables in a class statement will create the necessary indicator variables. The full stratified Cox PH model can be noted as follows

$$h_g(t) = h_{0g}(t)e^{\beta_1(\text{gender}) + \beta_2(\text{age}) + \beta_3(\text{male} * \text{treat}_A) + \beta_4(\text{gender} * \text{treat}_B) + \beta_5(\text{age} * \text{treat}_A) + \beta_6(\text{age} * \text{treat}_B)}$$

where each of the g equations {A, B, C} have varying slopes.

The reduced model can be noted as:

$$h_g(t) = h_{0g}(t)e^{\beta_1(\text{gender}) + \beta_2(\text{age})}$$

where each of the g equations {A, B, C} have varying slopes.

The slopes are allowed to be different for each treatment, A, B, and C, by plugging in different combinations of the parameters. The likelihood ratio test examines if the interactions terms are needed by testing the null hypothesis $\beta_3 = \beta_4 = \beta_5 = \beta_6 = 0$.

In SAS®, the likelihood ratio test for evaluating if the interaction model is needed for the stratified Cox PH can be completed in several steps. First, PROC PHREG must be run separately for both the full model (with the interaction(s) between the stratum variable and each covariate) and the reduced model (with no interaction(s)). The $-2\log L$ value for each model must be extracted from the PROC PHREG output by either manually looking at it in the automatic Model Fit output table or saving the output into a temporary dataset with either an `OUTSET=` option or an `ODS OUTPUT` option. A data step must then be used to compute the difference between these two $-2\log L$ values and generate the p-value with the `PROBCHI()` function, which returns the probability from a chi-square distribution. Our SAS® MACRO simplifies this by automating the above process for the user.

EXPLANATION OF THE MACRO CODE

The MACRO can be found and copied from the appendix of this paper. The most updated version of the MACRO is maintained in our GitHub repo at <https://github.com/SpectrumHealthResearch/lrt-strat-cox-ph>.

MACRO call:

```
%MACRO lrt_strat_cox_ph(data=, time_var=, censor_var=, censor_vals=,
  strata_vars=, quant_covariates=, class_covariates=, class_opts= );
```

Our SAS® MACRO has seven parameters to be defined by the user:

- data = the data set name
- time_var = the event time variable
- censor_var = the censoring indicator variable
- censor_vals = the value(s) for censored individuals
- strata_vars = the stratifying variable (the variable that does not meet the PH assumption)
 - Note: it MUST be categorical, so if it is quantitative, one must categorize it first
 - At this point, our MACRO only supports one stratifying variable, even though theoretically the stratified Cox PH model can include more than one stratifying variable.
- quant_covariates = the names of the numeric covariates in the model
- class_covariates = the names of the categorical covariates in the model
- class_opts = options for the class statement in PROC PHREG

Our SAS® MACRO contains two uses of PROC PHREG, which use the supplied dataset and specified variables. As seen in the code snippet below, the MACRO variable 'all_covariates' is created within the MACRO to include both the quantitative and categorical covariates specified by the user, and the local MACRO variable 'interaction_vars' is declared. A do loop is used to create all two-way interactions between each level of the stratum variable and each covariate.

```
%local error interaction_i interaction_vars all_covariates
%let all_covariates = &quant_covariates &class_covariates

%do interaction_i = 1 %to %sysfunc(countw(&all_covariates, %str( )));
  %let interaction_vars = &interaction_vars
```

Each use of PROC PHREG contains an ODS OUTPUT statement which saves the Type1 test output to temporary data sets called 'lrt_strat_coxph_type1_full' and 'lrt_strat_coxph_type1_red', respectively. These datasets contain the $-2\log L$ values from each model and the degrees of freedom (DF) associated with each. We are interested in the DF because each parameter in the full and reduced model is associated with 1 DF. In order to compute the DF for the likelihood ratio chi-square test, we need to know the difference in the number of parameters between each model. Thus, $DF_{full} - DF_{reduced}$ = the number of parameters in the full model – the number of parameters in the reduced model. This is seen in the MACRO code below:

```
proc phreg data=&data;
  class &class_covariates &strata_vars / &class_opts ;
  model &time_var*&censor_var(&censor_vals) = &all_covariates
    &interaction_vars / type1;
  strata &strata_vars;
  ods output Type1 = lrt_strat_cox_ph_type1_full;
run;

proc phreg data=&data;
  class &class_covariates &strata_vars / &class_opts ;
  model &time_var*&censor_var(&censor_vals) = &all_covariates / type1;
  strata &strata_vars;
  ods output Type1 = lrt_strat_cox_ph_type1_red;
run;
```

The first use of a data step in our MACRO sums the DF from the full model. The second data step sums the DF from the reduced model, and then computes the difference between the $-2\log L$ from each model

(which is the test statistic), as well as the DF, and uses the `PROBCHI()` function to generate the p-value of the likelihood ratio.

```
data lrt_strat_cox_ph_type1_full (keep=neg2ll_full df_full);
  set lrt_strat_cox_ph_type1_full (rename=(Neg2LogLike=neg2ll_full))
end=last;
  retain df_full 0;
  df_full = sum(df_full, DF);
  if last;
run;

data _null_;
  set lrt_strat_cox_ph_type1_red (rename=(Neg2LogLike=neg2ll_red))
end=last;
  retain df_red 0;
  df_red = sum(df_red, DF);
  if last then do;
    set lrt_strat_cox_ph_type1_full;
    diff = neg2ll_red - neg2ll_full;
    df = df_full - df_red;
    pvalue = 1-probchi(diff, df);
  end;
run;
```

EXAMPLE OF MACRO USE

The data set `addicts` has 238 heroin addicts as subjects and records their time in days from clinic entry until departure. It also includes a censoring variable, which of the two clinics the addicts are in, if they have a prison record, and their methadone dosage. The data may be obtained from <http://web1.sph.emory.edu/dkleinb/surv3.htm#data>.

Suppose we are interested in how methadone dose, prison record, and clinic affect the hazard rate of time until departure for heroin addicts. As shown in Figure 1, the variable `clinic` does not pass the PH assumption, and therefore cannot be an independent variable in the Cox PH model. However, we can stratify by `clinic` to incorporate it in the model.

MACRO Call:

```
%lrt_strat_cox_ph(
  data=addicts,
  time_var=survt,
  censor_var=status,
  censor_vals=0,
  strata_vars= clinic,
  quant_covariates= dose,
  class_covariates= prison,
  class_opts= param=glm
);
```

MACRO Output:

Table 1 Model Fit Statistics from the Full Model

Model Fit Statistics		
Criterion	Without Covariates	With Covariates
-2 LOG L	1229.367	1193.558
AIC	1229.367	1201.558
SBC	1229.367	1213.600

Table 2 Maximum Likelihood Estimates from the Full Model

Analysis of Maximum Likelihood Estimates								
Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
DOSE		1	-0.03691	0.01234	8.9425	0.0028	.	methadone dose (mg/day)
PRISON	0	1	0.08218	0.38430	0.0457	0.8307	.	0=none, 1=prison record 0
PRISON	1	0	0	0=none, 1=prison record 1
DOSE*CLINIC	1	1	0.00105	0.01457	0.0052	0.9427	.	Coded 1 or 2 1 * methadone dose (mg/day)
DOSE*CLINIC	2	0	0	Coded 1 or 2 2 * methadone dose (mg/day)
PRISON*CLINIC	0 1	1	-0.58467	0.42813	1.8650	0.1721	.	0=none, 1=prison record 0 * Coded 1 or 2 1
PRISON*CLINIC	0 2	0	0	0=none, 1=prison record 0 * Coded 1 or 2 2
PRISON*CLINIC	1 1	0	0	0=none, 1=prison record 1 * Coded 1 or 2 1
PRISON*CLINIC	1 2	0	0	0=none, 1=prison record 1 * Coded 1 or 2 2

Table 3 Type 1 Test Output from the Full Model

Likelihood Ratio Statistics for Type 1 Analysis				
Source	-2 Log L	DF	LR Chi-Square	Pr > ChiSq
(Without Covariates)	1229.3673	.	.	.
DOSE	1200.6656	1	28.7017	<.0001
PRISON	1195.4280	1	5.2376	0.0221
DOSE*CLINIC	1195.4279	1	0.0001	0.9918
PRISON*CLINIC	1193.5578	1	1.8701	0.1715

Table 4 Model Fit Statistics from the Reduced Model

Model Fit Statistics		
Criterion	Without Covariates	With Covariates
-2 LOG L	1229.367	1195.428
AIC	1229.367	1199.428
SBC	1229.367	1205.449

Table 5 Maximum Likelihood Estimates from the Reduced Model

Analysis of Maximum Likelihood Estimates								
Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
DOSE		1	-0.03514	0.00647	29.5471	<.0001	0.965	methadone dose (mg/day)
PRISON	0	1	-0.38877	0.16892	5.2974	0.0214	0.678	0=none, 1=prison record 0
PRISON	1	0	0	0=none, 1=prison record 1

Table 6 Type 1 Test Output from the Reduced Model

Likelihood Ratio Statistics for Type 1 Analysis				
Source	-2 Log L	DF	LR Chi-Square	Pr > ChiSq
(Without Covariates)	1229.3673	.	.	.
DOSE	1200.6656	1	28.7017	<.0001
PRISON	1195.4280	1	5.2376	0.0221

Output 1 Likelihood Ratio Test Output

Stratified Cox Proportional Hazards Model Likelihood Ratio Test Summary of results		
-2LogLikelihood of the Reduced Model	=	1195.43
-2LogLikelihood of the Full Model	=	1193.56
Degrees of Freedom of the Reduced Model	=	2
Degrees of Freedom of the Full Model	=	4
Model Degrees of Freedom	=	2
Difference	=	1.87
Chi-Square p-value	=	0.3925

Interpretation:

As shown in Table 7, the p-value of 0.3925 indicates there is not significant evidence to conclude the interaction parameters in the full model are not equal to zero. Thus, we may conclude that the full model is not significantly better than the reduced, and that the interaction between the stratified variable clinic and the two independent variables prison and dose is not needed.

CONCLUSION

Many statisticians use the likelihood ratio test to choose between a class of hierarchical models. In survival analysis, the likelihood ratio test can be used to determine whether interaction terms are needed for a stratified Cox PH Model. Our goal was to alleviate the burden of having to hard code or hand calculate the likelihood ratio test by using this SAS® MACRO, while still producing the information needed to determine if interaction terms should be included in a stratified Cox PH model.

REFERENCES

Gillespie, Brenda. "Checking Assumptions In The Cox Proportional Hazards Regression Model". *Mwsug.Org*, 2006, <https://www.mwsug.org/proceedings/2006/stats/MWSUG-2006-SD08.pdf>. Accessed 20 June 2019.

Kleinbaum, David G et al. *Applied Regression Analysis and Other Multivariate Methods*. 4th ed., Thompson Higher Education, 2009, pp.595-598

Kleinbaum, David G, and Mitchel Klein. *Survival Analysis*. 3rd ed., Springer New York, 2012.

Ying, Yao (2018). Several Methods to assess proportional hazard assumption when applying COX regression model. *Boehringer Ingelheim Inc., Shanghai, China, PharmaSUG China* 2018.

ACKNOWLEDGEMENTS

The entire Scientific Support Team at Spectrum Health Research for their feedback on our paper. A special thanks to Paul Egeler for his help in the development of our macro.

Dr. Robert Downer at Grand Valley State University for his encouragement to us to pursue submitting a paper. Also, Dr. Daniel Frobish at Grand Valley State University for inspiring us on this topic during his lectures in a first semester course on Survival Analysis and for his excellent critiques on our paper.

RECOMMENDED READINGS

- Kleinbaum, David G, and Mitchel Klein. *Survival Analysis*. 3rd ed., Springer New York, 2012.
- <https://github.com/SpectrumHealthResearch/lrt-strat-cox-ph>

CONTACT INFORMATION

Any comments or questions are valued and encouraged. You may contact the authors at:

Rachel R. Baxter
Grand Valley State University
Spectrum Health Office of Research
baxterac@mail.gvsu.edu

Katelyn J Ware
Grand Valley State University
Spectrum Health Office of Research
warek@mail.gvsu.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX

```

/*****
/** Stratified Cox Proportional Hazards Model Likelihood Ratio Test MACRO **/
*****/
/*
/* Authors: Rachel R. Baxter, B.S.; Katelyn J. Ware, B.A.;
/*          Paul W. Egeler, M.S., GStat
/*
/* Required parameters:
/*
/* data          = The data set name
/* time_var      = The event time variable
/* censor_var    = The censoring indicator variable
/* censor_vals   = The value(s) for censored individuals
/* strata_vars   = The stratifying variable (MUST BE CATEGORICAL)
/* quant_covariates = The names of numeric covariates in the model
/* class_covariates = The names of categorical covariates in the model
/* class_opts    = Options for the class statement in PROC PHREG
/*
*****/

%MACRO lrt_strat_cox_ph(
    data          = /* The data set name
    time_var      = /* The event time variable
    censor_var    = /* The censoring indicator variable
    censor_vals   = /* The value(s) for censored individuals
    strata_vars   = /* The stratifying variable (MUST BE CATEGORICAL)
    quant_covariates = /* The names of numeric covariates in the model
    class_covariates = /* The names of categorical covariates in the model
    class_opts    = /* Options for the class statement in PROC PHREG
);

/* Local variables */
%local error interaction_i interaction_vars all_covariates;
%let error = 0;
%let all_covariates = &quant_covariates &class_covariates;

/* User Input Processing */
%if %sysfunc(countw(&strata_vars, %str( ))) ne 1 %then %do;
    %put ERROR: strata_vars requires exactly one variable;
    %let error = 1;
%end;

%if &error = 1 %then %goto finish;

/* Create all pairwise interactions between strata_vars and covariates */
%do interaction_i = 1 %to %sysfunc(countw(&all_covariates, %str( )));
    %let interaction_vars = &interaction_vars &strata_vars *
%scan(&all_covariates, &interaction_i);
%end;

/* FULL MODEL */
proc phreg data=&data;
    class &class_covariates &strata_vars / &class_opts;
    model &time_var*&censor_var(&censor_value) = &all_covariates
        &interaction_vars / typel;

```

```

strata &strata_vars;
ods output Type1 = lrt_strat_cox_ph_type1_full;
run;

data lrt_strat_cox_ph_type1_full (keep=neg2ll_full df_full);
  set lrt_strat_cox_ph_type1_full (rename=(Neg2LogLike=neg2ll_full))end=last;
  retain df_full 0;
  df_full = sum(df_full, DF);
  if last;
run;

/* REDUCED MODEL */
proc phreg data=&data;
  class &class_covariates &strata_vars / &class_opts;
  model &time_var*&sensor_var(&sensor_value) = &all_covariates / type1;
  strata &strata_vars;
  ods output Type1 = lrt_strat_cox_ph_type1_red;
run;

/* Final processing and output results */
data _null_;
  set lrt_strat_cox_ph_type1_red (rename=(Neg2LogLike=neg2ll_red)) end=last;
  retain df_red 0;
  df_red = sum(df_red, DF);
  if last then do;
    set lrt_strat_cox_ph_type1_full;
    diff = neg2ll_red - neg2ll_full;
    df = df_full - df_red;
    pvalue = 1-probchi(diff, df);
    file print;
    HBAR1 = REPEAT("=", 80);
    HBAR2 = REPEAT("-", 80);
    put
      HBAR1
      / @5 "Stratified Cox Proportional Hazards Model Likelihood Ratio Test"
      / @25 "Summary of results"
      / HBAR2
      / @7 "-2LogLikelihood of the Reduced Model = " neg2ll_red 31.2
      / @7 "-2LogLikelihood of the Full Model = " neg2ll_full 31.2
      / @7 "Degrees of Freedom of the Reduced Model = " df_red 31.
      / @7 "Degrees of Freedom of the Full Model = " df_full 31.
      / @7 "Model Degrees of Freedom = " DF 31.
      / @7 "Difference = " diff 31.2
      / @7 "Chi-Square p-value = " pvalue 31.4
      / HBAR1;
  end;
run;

/* Clean up datasets used in MACRO */
proc datasets lib=work memtype=data noprint;
  delete lrt_strat_cox_ph_type1_red lrt_strat_cox_ph_type1_full;
  quit;
run;

%finish:

%mend lrt_strat_cox_ph;

```